


General Disclaimer

One or more of the Following Statements may affect this Document

- This document has been reproduced from the best copy furnished by the organizational source. It is being released in the interest of making available as much information as possible.
- This document may contain data, which exceeds the sheet parameters. It was furnished in this condition by the organizational source and is the best copy available.
- This document may contain tone-on-tone or color graphs, charts and/or pictures, which have been reproduced in black and white.
- This document is paginated as submitted by the original source.
- Portions of this document are not fully legible due to the historical nature of some of the material. However, it is the best reproduction available from the original submission.



SCHOOL OF ENGINEERING
OLD DOMINION UNIVERSITY
NORFOLK, VIRGINIA

Technical Report 76-T7

(NASA-CR-147116) EVALUATION OF
TRANSMITTANCE OF SELECTED INFRARED BANDS
(Old Dominion Univ., Norfolk, Va.) 36 p HC
\$4.00 CSCI 20F

N76-22991

Unclas
25284

G3/74

EVALUATION OF TRANSMITTANCE OF SELECTED INFRARED BANDS

By

S. K. Gupta

and

S. N. Tiwari

Prepared for the
National Aeronautics and Space Administration
Langley Research Center
Hampton, Virginia

Under
Grant NSG 1153

April 1976



SCHOOL OF ENGINEERING
OLD DOMINION UNIVERSITY
NORFOLK, VIRGINIA

Technical Report 76-T7

EVALUATION OF TRANSMITTANCE OF SELECTED INFRARED BANDS

By

S. K. Gupta

and

S. N. Tiwari

Prepared for the
National Aeronautics and Space Administration
Langley Research Center
Hampton, Virginia 23665

Under
Grant NSG 1153
Dr. Henry G. Reichle, Technical Monitor
Meteorology Section



Submitted by the
Old Dominion University Research Foundation
Norfolk, Virginia 23508

April 1976

FOREWORD

This report constitutes a part of the work done on the research project entitled "Radiative Transfer Models for Nonhomogeneous Atmosphere." The work was supported by the NASA Langley Research Center through Grant No. NSG-1153. The grant was monitored by Dr. Henry G. Reichle.

TABLE OF CONTENTS

	Page
FOREWORD	11
TABLE OF CONTENTS	111
LIST OF FIGURES	iv
SUMMARY	1
1. INTRODUCTION	2
2. SPECTRAL TRANSMITTANCE MODELS	3
2.1 Line-by-Line Model (Direct Integration)	3
2.2 Quasi-Random Band Model	5
3. TRANSMITTANCES OF INDIVIDUAL BANDS	6
3.1 CO 4.6 μ Band	6
3.2 N ₂ O 4.5 μ Band	8
3.3 CO ₂ 4.3 μ Band	8
3.4 H ₂ O 6.3 μ Band	11
3.5 CO ₂ 15 μ Band	16
4. CONCLUSIONS	19
REFERENCES	20
APPENDIX A - EXPLANATION OF SYMBOLS USED IN THE COMPUTER PROGRAMS	21
APPENDIX B - LINE-BY-LINE AND QUASI-RANDOM BAND MODEL COMPUTER PROGRAMS	25
B-1 Program GASLINE	25
B-2 Program GASBAND	29

LIST OF FIGURES

FIGURE NO.	TITLE
3.1	Comparison of transmittances of CO fundamental band.
3.2	Comparison of transmittances of 4.5μ N ₂ O band.
3.3	Comparison of transmittances of 4.3μ CO ₂ band.
3.4	Comparison of transmittances of 6.3μ H ₂ O band.
3.5	Comparison of transmittances of 15μ CO ₂ band.

EVALUATION OF TRANSMITTANCE OF
SELECTED INFRARED BANDS

by

S. K. Gupta and S. N. Tiwari

School of Engineering
Old Dominion University
Norfolk, Virginia 23508

SUMMARY

Computer programs have been developed for evaluating homogeneous path transmittance with line-by-line and quasi-random band model formulations. Transmittances for some selected bands of different gases have been obtained using these programs. Selected bands include the CO fundamental band (4.6μ), N_2O 4.5μ band, CO_2 4.3μ band, H_2O 6.3μ band and the CO_2 15μ band. Results of theoretical computations are compared with available experimental measurements. Significant errors are observed in the results obtained from the quasi-random band model formulation, indicating that it is inadequate to meet the accuracy requirements for atmospheric work.

1. INTRODUCTION

The work discussed in the present report developed from our studies of the application of radiative transfer models to the detection of gaseous pollutants of the atmosphere by remote sensing. It has been shown earlier [1]* and is further affirmed during the present investigation that the quasi-random band model computation of gas column transmittances is far more economical than the line-by-line model computation. It was considered important, therefore, to examine the possibility of using the quasi-random band model formulation for transmittance computations as required in surface temperature retrieval [2] and other data reduction procedures.

Transmittance computations were made for several bands of different gases using line-by-line and quasi-random band model formulations under conditions of temperature and pressure for which experimental measurements were available. Theoretical results are compared with the experimental results of reference [3]. In most cases, initial computations were made with higher resolution and the spectra were then degraded with appropriate slit functions for comparison with experimental results.

Spectral parameters for the lines of the bands considered in the present study were obtained from the AFCRL compilation [4,5] which is available at the Langley Research Center.

The spectral transmittance models are discussed briefly in Sec.2. Detailed comparison of the transmittances of the individual bands is presented in Sec.3.

*The numbers in brackets indicate references.

2. SPECTRAL TRANSMITTANCE MODELS

The spectral transmittance models (line-by-line and quasi-random band model) and the computation procedures used in the present work have been described in detail in [1]. These are discussed here briefly simply to direct attention to specific areas of present interest and to emphasize areas of important differences.

2.1 Line-by-Line Model (Direct Integration)

Drayson [6] and Kunde and Maguire [7] have described a procedure for computing the atmospheric transmittance by using the line-by-line model. The same procedure, with certain modifications described in [1], has been adapted in the present work. The entire frequency range of interest is first divided into a large number of narrow intervals $\Delta\omega$. The width $\Delta\omega$ is denoted in the program by DEL. Each interval is then divided into a variable number of subintervals depending upon the number of lines within the interval. Two very narrow subintervals are created on each side of a line center. The width of each of these subintervals is denoted by ALX in the program. The transmittance is computed at four frequency locations in each subinterval and is finally averaged over each interval.

The monochromatic transmittance at any wavenumber location ω is given by

$$\tau(\omega) = \exp[-\kappa(\omega) u] , \quad (2.1)$$

where $\kappa(\omega)$ is the total absorption coefficient at the wavenumber ω in $\text{cm}^{-1} - \text{atm}^{-1}$ and u represents the pressure path-length of the absorber in cm-atm .

Total absorption coefficient at any wavenumber ω consists of

contributions from all the lines in the vicinity and is computed in two parts as

$$\kappa(\omega) = \kappa^D(\omega) + \kappa^W(\omega) , \quad (2.2)$$

where $\kappa^D(\omega)$ and $\kappa^W(\omega)$ are called the direct and wing contributions respectively. Direct contribution originates from lines in very close vicinity (on both sides) and is obtained from

$$\kappa^D(\omega) = \sum_n S_n \gamma_n / \{ \pi [(\omega - \omega_n)^2 + \gamma_n^2] \} , \quad (2.3)$$

where ω_n refers to the center of the nth contributing line. In the computer program, the range of direct contribution is denoted by DLIM. Thus, if $|(\omega - \omega_n)| \leq \text{DLIM}$, the contribution is called direct and is evaluated by using Eq. (2.3). The summation in this equation extends over all lines from minus DLIM to plus DLIM. The wing contribution arises from lines which are farther from ω than DLIM and is obtained from

$$\kappa^W(\omega) = \sum_n S_n \gamma_n / [\pi (\omega - \omega_n)^2] . \quad (2.4)$$

The wave number range of the wing contribution is denoted by WLIM such that $\text{DLIM} < |(\omega - \omega_n)| \leq \text{WLIM}$. Equations (2.3) and (2.4) are based on the assumption that the line shapes are Lorentzian at the tropospheric pressures.

Different values of the parameters DEL and ALX have been chosen for different bands depending on the frequency range of the band and the gas pressure. The line-by-line computer program is given the name GASLINE. Various symbols used in the program are explained in Appendix A and a listing of the program is given in Appendix B-1.

2.2 Quasi-Random Band Model

In this formulation, the frequency range of the band is divided into intervals ($\Delta\omega$) of equal width which is denoted in the computer program by DEL. The lines within each interval are distributed into five intensity decades and average intensity for each decade is obtained. Average transmittance over the interval $\Delta\omega$, due to a single line of intensity S , is given by [8]

$$\bar{\tau}(\Delta\omega) = \frac{1}{\Delta\omega} \int_{\Delta\omega} \exp[-S u b(\omega, \omega_0)] d\omega_0, \quad (2.5)$$

where again u represents the pressure path-length in cm-atm and $b(\omega, \omega_0)$ is the shape factor for the line with center at ω_0 .

Expression (2.5) can easily be extended, first to include the contribution from all the lines within one intensity decade and then from the entire interval [8,9]. In this case also, it is important to take into account the contribution from lines in some adjacent intervals. The number of adjacent intervals (on both sides), from which the contribution is considered significant, is denoted by JD in the program. Different values of DEL have been adopted for different bands. The computer program is given the name GASBAND. The symbols used in the program are explained in Appendix A and a listing of the program is given in Appendix B-2.

3. TRANSMITTANCES OF INDIVIDUAL BANDS

The bands chosen for the comparison of transmittances in the present study are the CO fundamental band and those bands of other minor atmospheric constituents which interfere spectrally with the CO fundamental band. The CO₂ 15 μ band has been chosen because of its importance in various atmospheric radiative transfer analyses. Transmittance results are presented in Figs. 3.1-3.5 separately for each band. Line-by-line results are shown by the solid lines, quasi-random band model results by the histograms and the experimental results of Burch et al. [3] by the broken lines. Integrated absorptance for these bands are presented for comparison in Table 3.1.

3.1 CO Fundamental (4.6) Band

The frequency range selected for the study of this band (2070-2220 cm⁻¹) is the same as the range of operation of the gas-filter correlation instrument, which is flown from the Langley Research Center. Comparison for this band is shown in Fig. 3.1. A value of $DFL = 1 \text{ cm}^{-1}$ was used in the line-by-line computation. Due to low pressure of the gas in this case (51 mm Hg), resulting in small line widths, a value of $ALX = 0.003 \text{ cm}^{-1}$ was used. For the band model computation, a value for $DEL = 5 \text{ cm}^{-1}$ was used. The experimental results were obtained with an effective slit-width of 25 cm⁻¹ [3]. The theoretical results, therefore, were also degraded with a slit function of 25 cm⁻¹ for comparison. It can be seen from Fig. 3.1 that the line-by-line results are in good agreement with the experimental results while the band model results exhibit appreciable differences, particularly in the P and R branches of the band. Integrated absorptances for this band, presented in Table 3.1, are for slightly greater frequency range than previously considered. This was necessary because the experimentally measured absorptance covered a




Fig. 3.1 Comparison of transmittances of CO fundamental band.

greater frequency range. The results presented in Table 3.1 show that while the absorptance obtained by the line-by-line model is only five per cent lower than the experimental value, it is nine per cent lower for the band model absorptance.

3.2 N₂O 4.5 μ Band

The frequency range selected for this band is 2140-2290 cm^{-1} and the results are shown in Fig. 3.2. For the line-by-line computation, a value of $\text{DEL} = 1 \text{ cm}^{-1}$ was adopted. Since the gas pressure in this case was relatively high (644 mm Hg), $\text{ALX} = 0.02 \text{ cm}^{-1}$ was used. For the band model computation, values of $\text{DEL} = 5 \text{ cm}^{-1}$ and $\text{JD} = 10$ were adopted. The experimental results for this band were obtained for an effective slit-width of 15 cm^{-1} and, therefore, theoretical results were degraded with a slit function of 25 cm^{-1} for comparison. It is apparent from Fig. 3.2, that the experimental transmittance curve is shifted approximately 6 cm^{-1} to the higher frequency side with respect to the theoretical curves. The reason for this shift is not clear at this time. With the exception of this, the agreement between the line-by-line and experimental results is very good. The band model curve again shows appreciably low absorption. It can be seen from Table 3.1 that the line-by-line absorptance is less than two per cent lower than the experimental value while the difference for the band model absorptance is approximately ten per cent.

3.3 CO₂ 4.3 μ Band

The comparison of experimental and theoretical transmittances for this band, over the frequency range 2220-2420 cm^{-1} , is shown in Fig. 3.3. While the line-by-line results are obtained for $\text{DEL} = 1 \text{ cm}^{-1}$ and $\text{ALX} = 0.02 \text{ cm}^{-1}$, the band model results are for $\text{DEL} = 5 \text{ cm}^{-1}$ and $\text{JD} = 10$. Experimental results




Fig. 3.2 Comparison of transmittances of $4.5 \mu \text{ N}_2\text{O}$ band.

REPRODUCIBILITY OF THE
ORIGINAL PAGE IS POOR




Fig. 3.3 Comparison of transmittances of 4.3 μ CO₂ band.

shown were obtained for an effective slit-width of approximately 5 cm^{-1} . For purposes of comparison, therefore, the line-by-line results were degraded with a 5 cm^{-1} slit function while the band model results were only averaged over shifted and unshifted meshes. The agreement between the experimental and the line-by-line results is seen to be excellent. The band model curve again exhibits a slightly lower absorption. It can be seen from Table 3.1 that the integrated line-by-line absorptance is within 0.5 per cent of the experimental value while the band model absorptance is approximately 5.5 per cent lower.

3.4 H₂O 6.3 μ Band

The comparison for this band, which extends over the frequency region $1200 - 2100 \text{ cm}^{-1}$, is shown in Figs. 3.4a - 3.4c. It was found necessary to present these results in three figures to maintain clarity. For the frequency range of Fig. 3.4a, the experimental and line-by-line results are shown in Fig. 3.4(a-1) and the quasi-random band results in Fig. 3.4(a-2) again to maintain clarity.

The experimental results for this band were obtained with an effective slit-width of 6 cm^{-1} [3]. The present line-by-line results, which were obtained originally for $\text{DEL} = 2 \text{ cm}^{-1}$ and $\text{ALX} = 0.02 \text{ cm}^{-1}$ were degraded with a 6 cm^{-1} slit function. To ensure that no significant structure was lost, attempts were not made to smooth the results any further. Band model results were obtained for $\text{DEL} = 6 \text{ cm}^{-1}$ and $\text{JD} = 10$, and were only averaged over shifted and unshifted meshes. No further degradation of these was attempted.

Figures 3.4a show excellent agreement between the experimental and theoretical results in the lower frequency region of the band. In the middle and higher frequency regions (Figs. 3.4b and 3.4c), however, the theoretical




Fig. 3.4 (a-1) Comparison of transmittances of $6.3\mu \text{ H}_2\text{O}$ band ($\omega = 1200\text{-}1500 \text{ cm}^{-1}$).

REPRODUCIBILITY OF THE
ORIGINAL PAGE IS POOR




Fig. 3.4 (a-2) Comparison of transmittances of $6.3\mu \text{ H}_2\text{O}$ band ($1200\text{--}1500 \text{ cm}^{-1}$).




Fig. 3.4b Comparison of transmittances of $6.3\mu \text{ H}_2\text{O}$ band ($\omega = 1500\text{-}1800 \text{ cm}^{-1}$).




Fig. 3.4c Comparison of transmittances of $6.3\mu \text{ H}_2\text{O}$ band ($\omega = 1800\text{--}2100 \text{ cm}^{-1}$).

REPRODUCIBILITY OF THE
 ORIGINAL PAGE IS POOR

results exhibit much more structure than the experimental results. Integrated absorptances for this band, listed in Table 3.1, exhibit excellent agreement.

3.5 CO₂ 15 μ Band

Comparison of results for this band is shown in Fig. 3.5. The experimental results [3] refer to different values of effective slit-width over different parts of the band. The line-by-line results obtained for $DEL = 1 \text{ cm}^{-1}$ and $ALX = 0.02 \text{ cm}^{-1}$ were degraded by a 3 cm^{-1} slit function for the region $550\text{--}650 \text{ cm}^{-1}$ while a 5 cm^{-1} slit function was used for the region $650\text{--}800 \text{ cm}^{-1}$. The band model results were obtained for $DEL = 5 \text{ cm}^{-1}$ and $JD = 10$ and were averaged over shifted and unshifted meshes with no further degradation. It can be seen that the agreement between the three results is excellent. Table 3.1 shows that the theoretical results agree with the experimental ones within one per cent.

It is clear from the foregoing comparison that the line-by-line computation yields results in better agreement with the experimental values than the quasi-random band model. In view of the high accuracy required for atmospheric work, it is desirable to use the line-by-line model for the temperature retrieval work and other data reduction. Also, it is known that the spectra of the various atmospheric constituents under consideration have significant correlation among themselves. In the quasi-random band model, however, average transmittance over the interval is evaluated as a first step and so it is inherently unsuitable for computing transmittances of multi-component atmospheres.




Fig. 3.5 Comparison of transmittances of 15 μ CO₂ band.

Comparison of Integrated Absorptances

for Various Bands

TABLE 3.1

Band Identification and Freq. Range	Experimental Absorptance (Burch et al. 1962)	Line-by-Line Results		Band Model Results	
		Integrated Absorptance (cm^{-1})	Percentage Diff. with Measurement	Integrated Absorptance (cm^{-1})	Percentage Diff. with measurement
CO 4.6 μ (1970-2270 cm^{-1})	73.90	70.32	-4.84	67.30	-8.93
N ₂ O 4.5 μ (2140-2290 cm^{-1})	57.00	56.01	-1.74	51.43	-9.77
CO ₂ 4.3 μ (2220-2420 cm^{-1})	73.80	74.09	0.39	69.72	-5.53
H ₂ O 6.3 μ (1200-2100 cm^{-1})	334.0	326.4	-2.28	323.1	-3.26
CO ₂ 15 μ (550-800 cm^{-1})	66.10	66.57	0.71	65.35	-1.13

4. CONCLUSIONS

Computer programs have been developed for evaluating homogeneous path spectral transmittances with line-by-line and quasi-random band model formulations. The line-by-line program is given the name GASLINE and the band model the name GASBAND. By using these programs, transmittances for several bands of different gases (for selected pressure and path-length conditions) have been computed. The particular bands selected for this study were: CO 4.6μ , N₂O 4.5μ , CO₂ 4.3μ , H₂O 6.3μ , and CO₂ 15μ band. Theoretical results were compared with the experimental results of Burch et al. for the same set of pressure and path-length conditions. Comparisons of line-by-line and experimental results for transmittances indicate good agreement for all the bands. The quasi-random band model results for CO (4.6μ), N₂O (4.5μ), and CO₂ (4.3μ) bands, however, exhibit significant differences from the experimental results. Comparisons of integrated absorptance for these bands indicate the general trend observed for the transmittances.

REFERENCES

1. S. K. Gupta and S. N. Tiwari, "Evaluation of Upwelling Infrared Radiance from Earth's Atmosphere," Technical Report 75-T14, School of Engineering, Old Dominion University, Norfolk, Virginia (1975).
2. S. K. Gupta and S. N. Tiwari, "Retrieval of Surface Temperature by Remote Sensing," Technical Report 76-T8, School of Engineering, Old Dominion University, Norfolk, Virginia (1976).
3. D. E. Burch, D. A. Gryvnak, E. B. Singleton, W. L. France, and D. Williams, "Infrared Absorption by Carbon Dioxide, Water Vapor and Minor Atmospheric Constituents," AFCRL-62-698, Air Force Cambridge Research Laboratories, Bedford, Mass. (1962).
4. R. A. McClatchey, R. W. Fenn, J. E. A. Selby, F. E. Volz, J. S. Garing, "Optical Properties of the Atmosphere (Third Edition)," AFCRL-72-0497, Air Force Cambridge Research Laboratories, Bedford, Mass. (1972).
5. R. A. McClatchey, W. S. Benedict, S. A. Clough, D. E. Burch, R. F. Calfee, K. Fox, L. S. Rothman, and J. S. Garing, "AFCRL Atmospheric Line Parameters Compilation," AFCRL-TR-73-0096, Air Force Cambridge Research Laboratories, Bedford, Mass. (1973).
6. S. R. Drayson, Applied Optics 5, 385 (1966).
7. V. G. Kunde and W. C. Maguire, J. Quant. Spectrosc. Radiat. Transfer 14, 803 (1974).
8. P. J. Wyatt, V. R. Stull, and G. N. Plass, J. Opt. Soc. Amer. 52, 1209 (1962).
9. V. G. Kunde, "Theoretical Computations of the Outgoing Infrared Radiance from a Planetary Atmosphere," NASA TN D-4045, August 1967.

APPENDIX A

EXPLANATION OF SYMBOLS USED IN THE
COMPUTER PROGRAMSA-1. Symbols Used in Program GASLINE

AC	Monochromatic absorption coefficient at the specified frequency.
ACB, ACE & ACM }	Total wing contribution to the absorption coefficient at the interval boundaries and center.
AL	Half-widths of the individual lines.
ALX	Width of a narrow sub-interval near the line center.
DEL	Width of an interval.
DLIM	Width of the region from which the direct contribution is obtained.
EL	Energy of the lower states for the individual lines.
FR	Frequencies of the individual lines.
FRB	Frequencies at the interval boundaries.
FRC	Frequencies at the interval centers.
FRE	Frequencies of the lines falling within one interval.
FRG	Frequencies of the individual mesh points within one interval.
FRL	Lower frequency boundary of the band.
FRS	Frequencies at the sub-interval boundaries.
FRT	Upper frequency boundary of the band.
KR	Number of intervals in the band.
LE	Total number of lines in the spectrum.

MP	Number of lines in one interval.
NP	Number of sub-interval boundaries.
NQ	Number of sub-intervals in one interval.
PL	Pressure path-length of the absorber.
PNTP	Pressure referring to NTP (760 mm. Hg).
PREC	Effective pressure of the absorber.
RP	Exponent to account for the temperature-dependence of the rotational partition function.
SI	Intensities of the individual lines.
SL1, SL2	Slopes for the wing contribution between interval boundaries and the center.
TEMC	Temperature of the absorber ($^{\circ}\text{K}$).
TEMR	Reference temperature for the molecular spectral parameters ($^{\circ}\text{K}$).
TNTP	Temperature referring to NTP (273°K).
TR	Monochromatic transmittance at the specified frequency.
TRA	Transmittance averaged over each interval.
TRAV	Transmittance averaged over wider intervals to account for the slit functions.
VP	Factor accounting for the temperature-dependence of the vibrational partition function.
WL1, WL2	Weighting factors for the Gauss-Legendre quadrature formula.
WLIM	Width of the region from which the wing contribution is obtained.
XL1, XL2	Abcissa values for the Gauss-Legendre quadrature formula.

A-2. Symbols Used in Program GASBAND*

AL	Average half-width of the lines of the molecule.
AVSI	Average of the intensities of the lines within one intensity decade.
BIG	Upper intensity limit for individual intensity decades.
BIGI	Intensity of the strongest line within one interval.
IR	Index which lets the transmittances be computed for the unshifted and shifted meshes.
JD	Number of adjacent intervals on both sides of an interval from which the contribution is taken into account.
LIB	Serial number of the first line in an interval.
LIE	Serial number of the last line in an interval.
NSI	Number of lines within one decade of an interval
RES	Value of the integral for one line - direct or wing contribution.
SSI	Sum of the intensities of all the lines within one decade of an interval.
T1	Weighting factors used in the numerical integration of direct contribution.
T2	Weighting factors used in the numerical integration of wing contribution.
TRA	Array of transmittance values containing results from unshifted- and shifted-mesh computations.
TRD	Contribution to the total transmittance from lines within one interval.

*As far as possible, same symbols have been used in the Program GASBAND as in GASLINE. Only the symbols which are defined differently for GASBAND are explained here.

TRG Array of transmittances averaged over the width of an interval for unshifted or shifted mesh calculation. Results averaged over the above two sets are stored again in the same array.

X1 Abscissa values for the numerical integration of the direct contribution.

X2 Abscissa values for the numerical integration of the wing contribution.

APPENDIX B

LINE-BY-LINE AND QUASI-RANDOM BAND
MODEL COMPUTER PROGRAMSB-1. Program GASLINE

```

PROGRAM GASLINE(INPUT,OUTPUT)
INTEGER X,Y
DIMENSION FR(4590),SI(4590),EL(4590),AL(4590),FRB(451),FRC(450),
/FRE(100),FRS(126),FRG(500),AC(500),TR(500),TRA(450),TRAV(450)
READ 10, FRL,FRU,DEL,DLIM,WLIM
READ 10, PREC,PNTT,TEMC,TEMR,TNTP
READ 11, XL1,XL2,WL1,WL2
READ 12, PL,VP,ALX,RP,LE
READ 13, (FR(X),SI(X),EL(X),AL(X),X=1,LE)
10 FORMAT(10F8.2)
11 FORMAT(8F10.6)
12 FORMAT(4F8.4,18)
13 FORMAT(F11.3,E13.4,F11.3,F5.3)
PI=3.14159
C LINE INTENSITIES AND HALF-WIDTHS ARE CONVERTED FROM REFERENCE TO
C AMBIENT CONDITIONS OF TEMPERATURE AND PRESSURE
CST=(SQRT(TEMR/TEMC))*PREC/PNTT
PART=VP*((TEMR/TEMC)**RP)*2.69E+19
FACT=1.439*(TEMC-TEMR)/(TEMC*TEMR)
DO 101 X=1,LE
AL(X)=CST*AL(X)
101 SI(X)=SI(X)*PART*EXP(EL(X)*FACT)
C ENTIRE FREQUENCY RANGE IS DIVIDED INTO INTERVALS AND FREQUENCIES
C AT INTERVAL BOUNDARIES AND CENTERS ARE DEFINED
DELA=0.5*DEL
RK=(FRU-FRL)/DEL+0.1
KR=RK
FRB(1)=FRL
DO 102 K=1,KR
FRB(K+1)=FRB(K)+DEL
102 FRC(K)=FRB(K)+DELA
ALY=2.*ALX
C LOOP BELOW GOES TO THE END OF THE PROGRAM - COVERS ALL STEPS
Y=0
DO 103 K=1,KR
C DETERMINES THE NUMBER OF LINES (MP) IN THE PARTICULAR INTERVAL
M=0
MP=M
106 IF (Y-LE) 104,100,100
104 Y=Y+1
IF(FR(Y)-FRB(K)) 106,107,107
107 IF(FR(Y)-FRB(K+1)) 108,105,105
108 M=M+1
FRE(M)=FR(Y)

```

```

      MP=M
      GO TO 106
105 Y=Y-1
100 CONTINUE
C   DIVIDES THE INTERVAL INTO SUB-INTERVALS AND UP-DATES THE NUMBER OF
C   SUB-INTERVAL BOUNDARIES NP
      N=1
      FRS(N)=FRB(K)
      IF (MP) 109,109,110
110 CONTINUE
      DO 111 M=1,MP
      DIF=FRE(M)-FRS(N)
      IF (DIF-ALX) 112,112,113
113 IF (DIF-ALY) 114,114,115
112 FRS(N+1)=FRE(M)
      N=N+1
      GO TO 116
114 FRS(N+1)=FRE(M)-ALX
      FRS(N+2)=FRE(M)
      N=N+2
      GO TO 116
115 FRS(N+1)=FRE(M)-ALY
      FRS(N+2)=FRE(M)-ALX
      FRS(N+3)=FRE(M)
      N=N+3
116 IF (M-MP) 117,118,118
117 DIF=FRE(M+1)-FRS(N)
      IF (DIF-ALX) 119,119,120
120 IF (DIF-ALY) 121,121,122
119 GO TO 111
121 FRS(N+1)=FRE(M)+ALX
      N=N+1
      GO TO 111
122 FRS(N+1)=FRE(M)+ALX
      FRS(N+2)=FRE(M)+ALY
      N=N+2
111 CONTINUE
118 DIF=FRB(K+1)-FRE(M)
      IF (DIF-ALX) 123,123,124
124 IF (DIF-ALY) 125,125,126
123 FRS(N+1)=FRB(K+1)
      NP=N+1
      GO TO 127
125 FRS(N+1)=FRE(M)+ALX

```

```

      FRS(N+2)=FRB(K+1)
      NP=N+2
      GO TO 127
126  FRS(N+1)=FRE(M)+ALX
      FRS(N+2)=FRE(M)+ALY
      FRS(N+3)=FRB(K+1)
      NP=N+3
      GO TO 127
109  FRS(N+1)=FRB(K+1)
      NP=N+1
127  NQ=NP-1
C    GENERATES THE MESH BY COMPUTING FOUR FREQUENCY LOCATIONS IN EACH
C    SUB-INTERVAL
      DO 128 N=1,NQ
        VAR=0.5*(FRS(N+1)-FRS(N))
        CON=0.5*(FRS(N+1)+FRS(N))
        I=4*(N-1)+1
        FRG(I)=CON-VAR*XL1
        FRG(I+1)=CON-VAR*XL2
        FRG(I+2)=CON+VAR*XL2
128  FRG(I+3)=CON+VAR*XL1
        IG=4*NQ
C    INITIALIZES ABSORPTION COEFFICIENTS AT THE MESH POINTS AND THE
C    BOUNDARIES AND CENTERS OF THE INTERVAL
        ACB=0.
        ACM=0.
        ACE=0.
        DO 129 I=1,IG
129  AC(I)=0.
C    STARTS EVALUATION OF THE ABSORPTION COEFFICIENT
        DO 130 X=1,LE
          DIF=ABS(FR(X)-FRC(K))
          IF (DIF-WLIM) 131,131,130
131  IF (DIF-DLIM) 132,132,133
C    EVALUATES DIRECT CONTRIBUTION AT THE MESH POINTS
132  DO 134 I=1,IG
        FD=FR(X)-FRG(I)
        DEN=PI*(FD*FD+AL(X)*AL(X))
134  AC(I)=AC(I)+SI(X)*AL(X)/DEN
        GO TO 130
C    EVALUATES WING CONTRIBUTION AT INTERVAL BOUNDARIES AND CENTER
133  FB=FR(X)-FRB(K)
        FM=FR(X)-FRC(K)
        FE=FR(X)-FRB(K+1)

```

```

      PNUM=SI(X)*AL(X)
      ACB=ACB+PNUM/(PI*FB*FB)
      ACM=ACM+PNUM/(PI*FM*FM)
      ACE=ACE+PNUM/(PI*FE*FE)
130  CONTINUE
C    COMPUTES SLOPES FOR WING CONTRIBUTION BETWEEN INTERVAL
C    BOUNDARIES AND CENTER
      SL1=(ACM-ACB)/(FRC(K)-FRB(K))
      SL2=(ACE-ACM)/(FRB(K+1)-FRC(K))
C    EVALUATES TOTAL ABSORPTION COEFFICIENT AT ALL MESH POINTS
      DO 136 I=1,IG
        DIF=FRG(I)-FRB(K)
        IF (DIF-DELA) 137,138,138
137    AC(I)=AC(I)+ACB+SL1*DIF
        GO TO 136
138    AC(I)=AC(I)+ACM+SL2*(DIF-DELA)
136  CONTINUE
C    COMPUTES MONOCHROMATIC TRANSMITTANCES AT ALL MESH POINTS
      DO 141 I=1,IG
        OD=PL*AC(I)
        IF (OD.GT.675.) GO TO 153
        TR(I)=EXP(-OD)
        GO TO 141
153  TR(I)=0.
141  CONTINUE
C    EVALUATES TRANSMITTANCES AVERAGED OVER AN INTERVAL
      TRA(K)=0.
      DO 144 N=1,NQ
        VAR=0.5*(FRS(N+1)-FRS(N))
        I=4*(N-1)+1
        SUM1=TR(I)+TR(I+3)
        SUM2=TR(I+1)+TR(I+2)
        SUM=WL1*SUM1+WL2*SUM2
144  TRA(K)=TRA(K)+SUM*VAR/DEL
103  CONTINUE
C    SIMULATES SLIT FUNCTION TREATMENT BY TAKING 5-POINT SLIDING AVERAGES
      KN=KR-2
      DO 151 K=3,KN
151  TRAV(K)=(TRA(K-2)+TRA(K-1)+TRA(K)+TRA(K+1)+TRA(K+2))/5.
      PRINT 60, (TRA(K),K=1,KR)
      PRINT 60, (TRAV(K),K=3,KN)
60  FORMAT(1H1////////(10F10.5,/))
      STOP
      END

```


REPRODUCIBILITY OF THE
ORIGINAL PAGE IS POOR

B-2. Program GASBAND

```

PROGRAM GASBAND(INPUT,OUTPUT)
  INTEGER X,W
  DIMENSION FR(4590),SI(4590),EL(4590),FRB(152),FRC(151),LIB(151),
  /LIE(151),MP(151),BIGI(151),BIG(6,151),SSI(5,151),NSI(5,151),
  /AVSI(5,151),TRG(151),TRA(301),X1(26),T1(26),X2(21),T2(21)
  READ 10, FRL,FRU,DEL,PNT,PREC,TNTP,TEMC,TEMR
  READ 11, LE,JD
  READ 12, PL,VP,RP
  READ 12, (X1(W),W=1,26)
  READ 12, (T1(W),W=1,26)
  READ 12, (X2(W),W=1,21)
  READ 12, (T2(W),W=1,21)
  READ 13, (FR(X),SI(X),EL(X),X=1,LE)
10 FORMAT(10F8.1)
11 FORMAT(16I5)
12 FORMAT(10F8.4)
13 FORMAT(2(F11.3,E13.4,F11.3,5X))
C   CONVERTS THE LINE INTENSITIES AND THE AVERAGED LINE-WIDTHS FROM
C   REFERENCE TO AMBIENT CONDITIONS OF TEMPERATURE AND PRESSURE
  CST=(SQRT(TEMR/TEMC))*PREC/PNT
  PART=VP*((TEMR/TEMC)**RP)*2.69E+19
  FACT=1.439*(TEMC-TEMR)/(TEMC*TEMR)
  DO 101 X=1,LE
101 SI(X)=SI(X)*PART*EXP(EL(X)*FACT)
  ALA=AL*CST
C   INITIALIZES THE FREQUENCY INTERVAL BOUNDARIES FOR THE UNSHIFTED MESH
  DELA=0.5*DEL
  RK=(FRU-FRL)/DEL+0.1
  KR=RK
  FRB(1)=FRL
  IR=0
C   STARTS THE COMPUTATION FOR THE UNSHIFTED MESH
  GO TO 121
122 KR=KR-1
  FRB(1)=FRL-DELA
C   DEFINES THE FREQUENCIES AT THE INTERVAL BOUNDARIES AND CENTERS
121 DO 100 K=1,KR
  FRB(K+1)=FRB(K)+DEL
100 FRC(K)=FRB(K)+DELA
C   DETERMINES THE NUMBER OF LINES IN EACH FREQUENCY INTERVAL
  X=0
  DO 102 K=1,KR
  M=0
104 IF (X.GE.LE) GO TO 103

```

```

      X=X+1
      IF (FR(X).LT.FRB(K)) GO TO 104
      IF (FR(X).GE.FRB(K+1)) GO TO 105
      M=M+1
      GO TO 104
105 IF (K.GE.KR) GO TO 103
      X=X-1
103 MP(K)=M
102 CONTINUE
C   ASSIGNS NUMBERS TO THE FIRST AND LAST LINE OF EACH INTERVAL
      X=0
      DO 106 K=1,KR
      IF (MP(K).EQ.0) GO TO 106
      LIB(K)=X+1
      LIE(K)=LIB(K)+MP(K)-1
      X=LIE(K)
106 CONTINUE
C   FINDS OUT THE HIGHEST INTENSITY IN THE INTERVAL, GENERATES THE
C   INTENSITY DECADES, DISTRIBUTES LINES INTO INTENSITY DECADES,
C   DETERMINES THE NUMBER OF LINES IN EACH DECADE AND THEN COMPUTES
C   AVERAGE INTENSITY FOR EACH DECADE
      DO 107 K=1,KR
      JB=LIB(K)
      JE=LIE(K)
      BIGI(K)=SI(JB)
      JC=JB+1
      DO 108 J=JC,JE
      IF (BIGI(K).GE.SI(J)) GO TO 108
      BIGI(K)=SI(J)
108 CONTINUE
      DO 109 I=1,5
      IX=-I+1
109 BIG(I,K)=BIGI(K)*10.**IX
      DO 110 I=1,5
      N=0
      SSI(I,K)=0.
      DO 111 J=JB,JE
      IF (SI(J).GT.BIG(I,K)) GO TO 111
      IF (SI(J).LE.BIG(I+1,K)) GO TO 111
      N=N+1
      SSI(I,K)=SSI(I,K)+SI(J)
      NSI(I,K)=N
111 CONTINUE
      IF (NSI(I,K).GE.1) GO TO 110

```

REPRODUCIBILITY OF THE
ORIGINAL PAGE

```

      NSI(I,K)=1
110 AVSI(I,K)=SSI(I,K)/NSI(I,K)
107 CONTINUE
      PI=3.14159
      RHO=ALA/DELA
C     STARTS THE COMPUTATION OF TRANSMITTANCES FOR EACH INTERVAL
      DO 112 K=1,KR
        TRG(K)=1.
        DO 113 J=1,KR
          TRD=1.
          JA=ABS(J-K)
          IF (JA.GT.JD) GO TO 113
          ZI=FRC(K)-FRC(J)
          EPSI=ZI/DELA
          DO 115 I=1,5
            NSJ=NSI(I,J)
            PNUM=RHO*RHO*AVSI(I,J)*PL/(PI*ALA)
            RES=0.
            IF (J.NE.K) GO TO 116
C          EVALUATES THE INTEGRAL FOR DIRECT CONTRIBUTION
            DO 117 W=1,26
              YY=PNUM/(X1(W)*X1(W)+RHO*RHO)
              IF (YY.GT.675.) GO TO 119
              Y=EXP(-YY)
              GO TO 117
            119 Y=0.
            117 RES=RES+Y*T1(W)
            GO TO 115
C          EVALUATES THE INTEGRAL FOR WING CONTRIBUTION
            116 DO 118 W=1,21
              YY=PNUM/((EPSI-X2(W))*(EPSI-X2(W)))
              IF (YY.GT.675.) GO TO 120
              Y=EXP(-YY)
              GO TO 118
            120 Y=0.
            118 RES=RES+Y*T2(W)
            RES=RES/6.
            115 TRD=TRD*RES**NSJ
            113 TRG(K)=TRG(K)*TRD
            112 CONTINUE
            IR=IR+1
            IF (IR.GT.1) GO TO 123
            DO 114 K=1,KR
              114 TRA(2*K-1)=TRG(K)
C          GOES TO 122 AND STARTS COMPUTATION FOR THE SHIFTED MESH
            GO TO 122
            123 DO 124 K=1,KR
              124 TRA(2*K)=TRG(K)
C          AVERAGES RESULTS OVER SHIFTED AND UNSHIFTED MESHES
            DO 125 K=1,KR
              125 TRG(K)=(TRA(2*K-1)+TRA(2*K)+TRA(2*K+1))/3.
              PRINT 60, (TRG(K),K=1,KR)
            60 FORMAT(1H1////(10F12.5,///))
            STOP
            END

```